

Compliance is Mandated by Federal Regulations Please Read Carefully

This document may contain a “Destination Control Statement” (DCS) or other label that limits or restricts the release of its contents. If there is no DCS on the document that addresses export compliance, then it may only be released to, reviewed by or transferred to:

- U.S. Citizens working on behalf of a U.S. agency or company with a NASA/JSC affiliation;
or
- a “Lawful Permanent Resident” (also called a “Permanent Resident Alien”) holding valid documentation (referred to as a “Green Card”) and working on behalf of a U.S. agency or company with a NASA/JSC affiliation

This document may not be transferred or released in any form to any non-U.S. person until the document has been reviewed for export compliance issues and an appropriate DCS has been included. This document shall not be posted electronically, transferred, or otherwise made available without prior approval by an authorized NASA official. Public release of NASA technical information is controlled and requires approval through JSC Form 548 or NASA Form 1676.

Contact your authorized Export Control official for more information.

Verify that this is the current/correct version before use.

This document is not to be posted electronically.

Scientific and Technical Information Center

Main Library/IS23
NASA/Johnson Space Center
2101 NASA Parkway
Houston, TX 77058-3696
281-483-4245
Jsc-sticnter@mail.nasa.gov
<http://stic.jsc.nasa.gov>

This page is not part of the original file.

A Fast Recursive Singularity Free Algorithm for Calculating the First and Second Derivatives of the Geopotential

Engineering Directorate
Navigation, Control, and Aeronautics
Division

July 6, 1990



National Aeronautics and
Space Administration

Lyndon B. Johnson Space Center
Houston, Texas


TECHNICAL LIBRARY
BUILDING 45
Johnson Space Center
Houston, Texas 77058


SHUTTLE PROGRAM

A Fast Recursive Singularity Free Algorithm for Calculating the First and Second Derivatives of the Geopotential

By: Robert G. Gottlieb, McDonnell Douglas

JSC Task Monitor: T. J. Blucker, Navigation Section

Approved: 
G. C. Hite, Chief
Navigation and Guidance Systems Branch

Approved: 
K. J. Cox, Acting Chief
Navigation, Control, and Aeronautics Division

Navigation, Control, and Aeronautics Division

National Aeronautics and Space Administration
Lyndon B. Johnson Space Center
Houston, Texas

July 6, 1990

CONTENTS

Section		Page
1	<u>SUMMARY</u>	1
2	<u>INTRODUCTION</u>	2
3	<u>THE FIRST PARTIAL</u>	5
4	<u>THE SECOND PARTIAL</u>	8
5	<u>NUMERICAL VERIFICATION</u>	14
6	<u>CONCLUSIONS</u>	15
7	<u>REFERENCES</u>	16
APPENDIX	<u>COMPUTER LISTINGS</u>	18

TABLES

Table		Page
I	NUMERICAL VERIFICATION OF RESULTS	17

SECTION 1

SUMMARY

This report presents the development of a recursive, non-singular method for computing the first and second partials of the geopotential with respect to the position vector, $\underline{X}^T = (x_1 \ x_2 \ x_3)$. The magnitude of \underline{X} is the radius, r .

The second partial, which is the partial derivative of the gravitational acceleration with respect to the position, is required in a number of disciplines; e.g., estimation, optimization, gravity gradient, etc.

A non-singular, recursive algorithm for computing the first and a symmetric, non-singular, recursive algorithm for computing the second derivative of the geopotential are derived, programmed, and verified.

SECTION 2 INTRODUCTION

This report presents the development of a recursive, non-singular method for computing the first and second partials of the geopotential with respect to the position vector, $\underline{X}^T = (x_1 \ x_2 \ x_3)$. The magnitude of \underline{X} is the radius, r .

The second partial, which is the partial derivative of the gravitational acceleration with respect to the position, is required by a number of disciplines; e.g., estimation, optimization, gravity gradient, etc.

The gravitational potential function for the Earth is normally written

$$V = -\frac{\mu}{r} - \sum_{n=2}^{\infty} \sum_{m=0}^n \frac{\mu}{r} \left(\frac{a_e}{r}\right)^n P_{n,m}(\epsilon) (C_{n,m} \cos m\lambda + S_{n,m} \sin m\lambda) \quad (1)$$

where μ is the gravitational constant, a_e is the equatorial radius of the Earth, and

$$P_{n,m} = (1 - \epsilon^2)^{m/2} \frac{\partial^m P_n}{\partial \epsilon^m}$$

are the associated Legendre functions and P_n are the Legendre polynomials. Also, we have the sine of the latitude

$$\epsilon = x_3/r$$

and the longitude is computed from

$$\lambda = \tan^{-1}(x_2/x_1)$$

For notational convention, we define a geopotential function U to be

$$U \equiv -V$$

and write

$$U = \frac{\mu}{r} + \sum_{n=2}^{\infty} \sum_{m=0}^n \frac{\mu}{r} \left(\frac{a_e}{r}\right)^n P_{n,m}(\epsilon) (C_{n,m} \cos m\lambda + S_{n,m} \sin m\lambda) \quad (2)$$

$P_{n,m}$ becomes

$$P_{n,m} = \left(\frac{r^2 - x_3^2}{r^2}\right)^{m/2} \frac{\partial^m P_n}{\partial \epsilon^m} = \frac{\rho^m}{r^m} P_n^m$$

where

$$\rho^2 = x_1^2 + x_2^2$$

and

$$P_n^m \equiv \frac{\partial^m P_n}{\partial \varepsilon^m}$$

Also, define

$$C_m \equiv \rho^m \cos m \lambda$$

$$S_m \equiv \rho^m \sin m \lambda$$

$$B_{n,m} \equiv C_{n,m} C_m + S_{n,m} S_m$$

The geopotential can now be written

$$U = \frac{\mu}{r} + \sum_{n=2}^{\infty} \sum_{m=0}^n \frac{\mu}{r} \left(\frac{a_e}{r} \right)^n \frac{P_n^m(\varepsilon)}{r^m} B_{n,m} \quad (3)$$

This form is especially useful since P_n^m , C_m , and S_m can be calculated recursively and the singularity at the pole ($\rho = 0$) can be avoided.

The Legendre polynomials may be calculated recursively as

$$P_n^m = P_{n-2}^m + (2n-1)P_{n-1}^{m-1}, \quad m \geq 1$$

$$P_n^0 = P_n = \left((2n-1)\varepsilon P_{n-1} - (n-1)P_{n-2} \right) / n$$

$$P_0^0 = 1, \quad P_0^1 = 0$$

$$P_1^0 = \varepsilon, \quad P_1^1 = 1$$

also note that

$$C_m = \rho^m \cos m \lambda = \rho^{m-1+1} \cos(m-1+1)\lambda = C_1 C_{m-1} - S_1 S_{m-1}$$

$$S_m = \rho^m \sin m \lambda = \rho^{m-1+1} \sin(m-1+1)\lambda = S_1 C_{m-1} + C_1 S_{m-1}$$

$$C_1 = \rho \cos \lambda = \rho \frac{x_1}{\rho} = x_1$$

$$S_1 = \rho \sin \lambda = \rho \frac{x_2}{\rho} = x_2$$

It is helpful during coding to note that $\tilde{C}_m \equiv C_m/r^m$ and $\tilde{S}_m \equiv S_m/r^m$ are also recursive, since

$$\tilde{C}_m = \frac{C_m}{r^m} = \frac{C_m}{r^{m-1+1}} = \tilde{C}_1 \tilde{C}_{m-1} - \tilde{S}_1 \tilde{S}_{m-1}$$

$$\tilde{S}_m = \frac{S_m}{r^m} = \frac{S_m}{r^{m-1+1}} = \tilde{S}_1 \tilde{C}_{m-1} + \tilde{C}_1 \tilde{S}_{m-1}$$

$$\tilde{C}_1 = \frac{x_1}{r} \quad , \quad \tilde{S}_1 = \frac{x_2}{r}$$

SECTION 3 THE FIRST PARTIAL

The gravitational acceleration, \underline{g} , is calculated as the first partial derivative of U with respect to the Earth-fixed position vector, \underline{X} . From equation (3), we have

$$\underline{g} = \frac{\partial U}{\partial \underline{X}} = \frac{\partial U}{\partial r} \frac{\partial r}{\partial \underline{X}} + \frac{\partial U}{\partial \varepsilon} \frac{\partial \varepsilon}{\partial \underline{X}} + \frac{\partial U}{\partial B_{n,m}} \frac{\partial B_{n,m}}{\partial \underline{X}} \quad (4)$$

where

$$\frac{\partial U}{\partial r} = -\frac{\mu}{r^2} - \sum \sum \frac{\mu}{r^2} \left(\frac{a_e}{r} \right)^n \frac{(n+m+1)}{r^m} P_n^m B_{n,m} \quad \dagger$$

$$\frac{\partial U}{\partial \varepsilon} = \sum \sum \frac{\mu}{r} \left(\frac{a_e}{r} \right)^n P_n^{m+1} \frac{B_{n,m}}{r^m}$$

$$\frac{\partial U}{\partial B_{n,m}} = \sum \sum \frac{\mu}{r} \left(\frac{a_e}{r} \right)^n \frac{P_n^m}{r^m}$$

Using the recursive definition of C_m and S_m and the fact that

$$\tan \lambda = x_2 / x_1$$

and

$$\underline{X}^T = (x_1 \ x_2 \ x_3)$$

it can now be shown that ^[1]

$$\frac{\partial B_{n,m}}{\partial \underline{X}} = \begin{bmatrix} m(C_{n,m} C_{m-1} + S_{n,m} S_{m-1}) \\ -m(C_{n,m} S_{m-1} - S_{n,m} C_{m-1}) \\ 0 \end{bmatrix} \equiv m \begin{bmatrix} B_{n,m-1} \\ -A_{n,m-1} \\ 0 \end{bmatrix} \equiv m \underline{b}$$

[†]For notational simplicity, $\sum \sum = \sum_{n=2}^{\infty} \sum_{m=0}^n$ throughout.

Also,

$$\frac{\partial r}{\partial \underline{X}} = \frac{\underline{X}}{r}$$

and,

$$\frac{\partial e}{\partial \underline{X}} = \frac{1}{r} a - \frac{x_3}{r^3} \underline{X}$$

where $\underline{a}^T = (0 \ 0 \ 1)$

Combining these partials and substituting into equation (4), we get

$$\begin{aligned} \frac{\partial U}{\partial \underline{X}} = & -\frac{\mu}{r^2} \left(1 + \sum \sum \left(\frac{a_e}{r} \right)^n (n+m+1) \frac{P_n^m B_{n,m}}{r^m} \right) \frac{\underline{X}}{r} \\ & + \frac{\mu}{r^2} \sum \sum \left(\frac{a_e}{r} \right)^m P_n^{m+1} \frac{B_{n,m}}{r^m} \underline{a} - \frac{\mu}{r^2} \sum \sum \left(\frac{a_e}{r} \right)^n P_n^{m+1} \frac{B_{n,m}}{r^m} \frac{x_3}{r} \frac{\underline{X}}{r} \\ & + \frac{\mu}{r^2} \sum \sum \left(\frac{a_e}{r} \right)^n \frac{P_n^m}{r^{m-1}} \begin{bmatrix} m(C_{n,m} C_{m-1} + S_{n,m} S_{m-1}) \\ -m(C_{n,m} S_{m-1} - S_{n,m} C_{m-1}) \\ 0 \end{bmatrix} \end{aligned}$$

Defining,

$$J_n \equiv \sum_{m=1}^n m \frac{P_n^m}{r^{m-1}} (C_{n,m} C_{m-1} + S_{n,m} S_{m-1})$$

$$K_n \equiv - \sum_{m=1}^n m \frac{P_n^m}{r^{m-1}} (C_{n,m} S_{m-1} - S_{n,m} C_{m-1})$$

$$\Gamma_n \equiv C_{n,0} (n+1) P_n^0 + \sum_{m=1}^n (1+n+m) \frac{P_n^m}{r^m} (C_{n,m} C_m + S_{n,m} S_m)$$

$$H_n \equiv C_{n,0} P_n^1 + \sum_{m=1}^n \frac{P_n^{m+1}}{r^m} (C_{n,m} C_m + S_{n,m} S_m)$$

$$H \equiv \sum_{n=2}^{\infty} \left(\frac{a_e}{r} \right)^n H_n$$

$$\Gamma \equiv 1 + \sum_{n=2}^{\infty} \left(\frac{a_e}{r} \right)^n \Gamma_n$$

$$J \equiv \sum_{n=2}^{\infty} \left(\frac{a_e}{r} \right)^n J_n$$

$$K \equiv \sum_{n=2}^{\infty} \left(\frac{a_e}{r} \right)^n K_n$$

$$\Lambda \equiv \Gamma + \frac{x_3 H}{r}$$

$$\hat{X} \equiv \frac{\underline{X}}{r}$$

we can write

$$\underline{g} \equiv \frac{\partial U}{\partial \underline{X}} = -\frac{\mu}{r^2} \left(\Lambda \hat{X} - \begin{bmatrix} J \\ K \\ H \end{bmatrix} \right) \quad (5)$$

Note that the final result for the first partial derivative of the geopotential is a rather simple, compact, vector equation.

SECTION 4
THE SECOND PARTIAL

Next we calculate

$$\frac{\partial^2 U}{\partial \underline{X}^2}$$

starting with equation (4), as it leads directly to a compact symmetric notation.

We have from equation (4)

$$\frac{\partial U}{\partial \underline{X}} = \frac{\partial U}{\partial r} \frac{\partial r}{\partial \underline{X}} + \frac{\partial U}{\partial \epsilon} \frac{\partial \epsilon}{\partial \underline{X}} + \frac{\partial U}{\partial B_{n,m}} \frac{\partial B_{n,m}}{\partial \underline{X}}$$

Thus,

$$\begin{aligned} \frac{\partial^2 U}{\partial \underline{X}^2} &= \frac{\partial^2 U}{\partial r^2} \frac{\partial r}{\partial \underline{X}} \frac{\partial r^T}{\partial \underline{X}} + \frac{\partial^2 U}{\partial \epsilon^2} \frac{\partial \epsilon}{\partial \underline{X}} \frac{\partial \epsilon^T}{\partial \underline{X}} + \frac{\partial^2 U}{\partial B_{n,m}^2} \frac{\partial B_{n,m}}{\partial \underline{X}} \frac{\partial B_{n,m}^T}{\partial \underline{X}} \\ &+ \frac{\partial^2 U}{\partial \epsilon \partial r} \left(\frac{\partial r}{\partial \underline{X}} \frac{\partial \epsilon^T}{\partial \underline{X}} + \frac{\partial \epsilon}{\partial \underline{X}} \frac{\partial r^T}{\partial \underline{X}} \right) + \frac{\partial^2 U}{\partial r \partial B_{n,m}} \left(\frac{\partial B_{n,m}}{\partial \underline{X}} \frac{\partial r^T}{\partial \underline{X}} + \frac{\partial r}{\partial \underline{X}} \frac{\partial B_{n,m}^T}{\partial \underline{X}} \right) \\ &+ \frac{\partial^2 U}{\partial \epsilon \partial B_{n,m}} \left(\frac{\partial B_{n,m}}{\partial \underline{X}} \frac{\partial \epsilon^T}{\partial \underline{X}} + \frac{\partial \epsilon}{\partial \underline{X}} \frac{\partial B_{n,m}^T}{\partial \underline{X}} \right) + \frac{\partial U}{\partial r} \frac{\partial^2 r}{\partial \underline{X}^2} + \frac{\partial U}{\partial \epsilon} \frac{\partial^2 \epsilon}{\partial \underline{X}^2} + \frac{\partial U}{\partial B_{n,m}} \frac{\partial^2 B_{n,m}}{\partial \underline{X}^2} \end{aligned} \quad (6)$$

The second and cross partials appearing in equation (6) are

$$\frac{\partial^2 U}{\partial r^2} = \frac{2\mu}{r^3} + \sum \sum \mu \left(\frac{a_e}{r} \right)^n \frac{(m+n+1)}{r^3} \frac{(m+n+2)}{r^m} P_n^m B_{n,m}$$

$$\frac{\partial^2 U}{\partial \epsilon^2} = \sum \sum \frac{\mu}{r} \left(\frac{a_e}{r} \right)^n \frac{1}{r^m} P_n^{m+2} B_{n,m}$$

$$\frac{\partial^2 U}{\partial B_{n,m}} = 0$$

$$\frac{\partial^2 U}{\partial \epsilon \partial r} = - \sum \sum \frac{\mu}{r^2} \left(\frac{a_e}{r} \right)^n \frac{(m+n+1)}{r^m} P_n^{m+1} B_{n,m}$$

$$\frac{\partial^2 U}{\partial \varepsilon \partial B_{n,m}} = \sum \sum \frac{\mu}{r} \left(\frac{a_e}{r} \right)^n \frac{P_n^{m+1}}{r^m}$$

$$\frac{\partial^2 U}{\partial r \partial B_{n,m}} = - \sum \sum \frac{\mu}{r^2} \left(\frac{a_e}{r} \right)^n \frac{(m+n+1)}{r^m} P_n^m$$

$$\frac{\partial r}{\partial \underline{X}} \frac{\partial r^T}{\partial \underline{X}} = \frac{\underline{X} \underline{X}^T}{r^2}$$

$$\frac{\partial \varepsilon}{\partial \underline{X}} \frac{\partial \varepsilon^T}{\partial \underline{X}} = \frac{x_3^2}{r^6} \underline{X} \underline{X}^T - \frac{x_3}{r^4} (\underline{X} \underline{a}^T + \underline{a} \underline{X}^T) + \frac{1}{r^2} \underline{a} \underline{a}^T$$

$$\frac{\partial^2 r}{\partial \underline{X}^2} = \frac{1}{r} \left(1 - \frac{\underline{X} \underline{X}^T}{r^2} \right)$$

$$\frac{\partial^2 \varepsilon}{\partial \underline{X}^2} = \frac{3x_3}{r^5} \underline{X} \underline{X}^T - \frac{1}{r^3} (\underline{X} \underline{a}^T + \underline{a} \underline{X}^T) - \frac{x_3}{r^3} I$$

$$\frac{\partial^2 B_{n,m}}{\partial \underline{X}^2} = m(m-1) \begin{bmatrix} B_{n,m-2} & -A_{n,m-2} & 0 \\ -A_{n,m-2} & -B_{n,m-2} & 0 \\ 0 & 0 & 0 \end{bmatrix}$$

where

$$B_{n,m-2} \equiv C_{n,m} C_{m-2} + S_{n,m} S_{m-2}$$

$$A_{n,m-2} \equiv C_{n,m} S_{m-2} - S_{n,m} C_{m-2}$$

We also note the special combinations

$$\frac{\partial r}{\partial \underline{X}} \frac{\partial \varepsilon^T}{\partial \underline{X}} + \frac{\partial \varepsilon}{\partial \underline{X}} \frac{\partial r^T}{\partial \underline{X}} = \frac{1}{r^2} (\underline{X} \underline{a}^T + \underline{a} \underline{X}^T) - 2 \frac{x_3}{r^4} \underline{X} \underline{X}^T$$

and

$$\begin{aligned} \frac{\partial B_{n,m}}{\partial \underline{X}} \frac{\partial r^T}{\partial \underline{X}} + \frac{\partial r}{\partial \underline{X}} \frac{\partial B_{n,m}^T}{\partial \underline{X}} &= m \begin{bmatrix} B_{n,m-1} \\ -A_{n,m-1} \\ 0 \end{bmatrix} \frac{\underline{X}^T}{r} + m \frac{\underline{X}}{r} [B_{n,m-1} \quad -A_{n,m-1} \quad 0] \\ &= \frac{m}{r} (\underline{b} \underline{X}^T + \underline{X} \underline{b}^T) \end{aligned}$$

where

$$\underline{b} \equiv \begin{bmatrix} B_{n,m-1} \\ -A_{n,m-1} \\ 0 \end{bmatrix}$$

and

$$\frac{\partial B_{n,m}}{\partial \underline{X}} \frac{\partial \varepsilon^T}{\partial \underline{X}} + \frac{\partial \varepsilon}{\partial \underline{X}} \frac{\partial B_{n,m}^T}{\partial \underline{X}} = \frac{m}{r} \left[(\underline{b} \underline{a}^T + \underline{a} \underline{b}^T) - \frac{x_3}{r^2} (\underline{b} \underline{X}^T + \underline{X} \underline{b}^T) \right]$$

Putting all these into equation (6) leads to

$$\begin{aligned} \frac{\partial^2 U}{\partial \underline{X}^2} &= \frac{\mu}{r^3} \left\{ 2 + \sum \sum \left(\frac{a_e}{r} \right)^n (m+n+1) \frac{(m+n+2)}{r^m} P_n^m B_{n,m} \right\} \frac{\underline{X} \underline{X}^T}{r^2} \\ &+ \frac{\mu}{r^3} \sum \sum \left(\frac{a_e}{r} \right)^n P_n^{m+2} \frac{B_{n,m}}{r^m} \left[\frac{x_3^2}{r^4} \underline{X} \underline{X}^T \frac{x_3}{r^2} (\underline{X} \underline{a}^T + \underline{a} \underline{X}^T) + \underline{a} \underline{a}^T \right] \\ &- \frac{\mu}{r^3} \sum \sum \left(\frac{a_e}{r} \right)^n \frac{(n+m+1)}{r^m} P_n^{m+1} B_{n,m} \left[\frac{1}{r} (\underline{X} \underline{a}^T + \underline{a} \underline{X}^T) - 2 \frac{x_3}{r^3} \underline{X} \underline{X}^T \right] \\ &+ \frac{\mu}{r^3} \sum \sum \left(\frac{a_e}{r} \right)^n \frac{P_n^{m+1}}{r^{m-1}} m \left[(\underline{b} \underline{a}^T + \underline{a} \underline{b}^T) - \frac{x_3}{r^2} (\underline{b} \underline{X}^T + \underline{X} \underline{b}^T) \right] \\ &- \frac{\mu}{r^3} \sum \sum \left(\frac{a_e}{r} \right)^n \frac{(n+m+1)}{r^{m-1}} P_n^m \frac{m}{r} \left[(\underline{b} \underline{X}^T + \underline{X} \underline{b}^T) \right] \\ &- \frac{\mu}{r^3} \left(1 + \sum \sum \left(\frac{a_e}{r} \right)^n \frac{(n+m+1)}{r^m} P_n^m B_{n,m} \right) \left[I - \frac{\underline{X} \underline{X}^T}{r^2} \right] \\ &+ \frac{\mu}{r^3} \sum \sum \left(\frac{a_e}{r} \right)^n P_n^{m+1} \frac{B_{n,m}}{r^m} \left[3 \frac{x_3}{r^3} \underline{X} \underline{X}^T - \frac{1}{r} (\underline{X} \underline{a}^T + \underline{a} \underline{X}^T) - \frac{x_3}{r} I \right] \end{aligned}$$

$$+ \frac{\mu}{r^3} \sum \sum \left(\frac{a_e}{r} \right)^n \frac{P_n^m}{r^{m-2}} m(m-1) \begin{bmatrix} B_{n,m-2} & -A_{n,m-2} & 0 \\ -A_{n,m-2} & -B_{n,m-2} & 0 \\ 0 & 0 & 0 \end{bmatrix} \quad (7)$$

If we now define

$$L \equiv 2 + \sum_{n=2}^{\infty} \left(\frac{a_e}{r} \right)^n \sum_{m=0}^n (n+m+1)(n+m+2) P_n^m \frac{B_{n,m}}{r^m}$$

$$M \equiv \sum_{n=2}^{\infty} \left(\frac{a_e}{r} \right)^n \sum_{m=0}^n \frac{P_n^{m+2}}{r^m} B_{n,m}$$

$$N \equiv \sum_{n=2}^{\infty} \left(\frac{a_e}{r} \right)^n \sum_{m=2}^n P_n^m m(m-1) \frac{(C_{n,m} C_{m-2} + S_{n,m} S_{m-2})}{r^{m-2}}$$

$$\Omega \equiv \sum_{n=2}^{\infty} \left(\frac{a_e}{r} \right)^n \sum_{m=2}^n P_n^m m(m-1) \frac{(C_{n,m} S_{m-2} - S_{n,m} C_{m-2})}{r^{m-2}}$$

$$P \equiv \sum_{n=2}^{\infty} \left(\frac{a_e}{r} \right)^n \sum_{m=0}^n P_n^{m+1} (m+n+1) \frac{B_{n,m}}{r^m}$$

$$Q \equiv \sum_{n=2}^{\infty} \left(\frac{a_e}{r} \right)^n \sum_{m=1}^n P_n^{m+1} m \frac{(C_{n,m} C_{m-1} + S_{n,m} S_{m-1})}{r^{m-1}}$$

$$R \equiv - \sum_{n=2}^{\infty} \left(\frac{a_e}{r} \right)^n \sum_{m=1}^n P_n^{m+1} m \frac{(C_{n,m} S_{m-1} - S_{n,m} C_{m-1})}{r^{m-1}}$$

$$S \equiv \sum_{n=2}^{\infty} \left(\frac{a_e}{r} \right)^n \sum_{m=1}^n (m+n+1) P_n^m m \frac{(C_{n,m} S_{m-1} - S_{n,m} C_{m-1})}{r^{m-1}}$$

$$T \equiv - \sum_{n=2}^{\infty} \left(\frac{a_e}{r} \right)^n \sum_{m=1}^n (m+n+1) P_n^m m \frac{(C_{n,m} S_{m-1} - S_{n,m} C_{m-1})}{r^{m-1}}$$

then with the definitions

$$\underline{a} \equiv \begin{bmatrix} Q \\ R \\ 0 \end{bmatrix}$$

$$\underline{Y} \equiv \begin{bmatrix} S \\ T \\ 0 \end{bmatrix}$$

we can write equation (7) as

$$\begin{aligned} \frac{\partial^2 U}{\partial \underline{X}^2} = & \frac{\mu}{r^3} L \frac{\underline{X} \underline{X}^T}{r^2} + \frac{\mu}{r^3} M \left[\frac{x_3^2}{r^4} \underline{X} \underline{X}^T - \frac{x_3}{r^2} (\underline{X} \underline{a}^T + \underline{a} \underline{X}^T) + \underline{a} \underline{a}^T \right] \\ & - \frac{\mu}{r^3} P \left[\frac{1}{r} (\underline{X} \underline{a}^T + \underline{a} \underline{X}^T) - 2 \frac{x_3}{r^3} \underline{X} \underline{X}^T \right] + \frac{\mu}{r^3} \left[(\underline{a} \underline{a}^T + \underline{a} \underline{a}^T) - \frac{x_3}{r^3} (\underline{a} \underline{X}^T + \underline{X} \underline{a}^T) \right] \\ & - \frac{\mu}{r^3} \left[\frac{\underline{Y} \underline{X}^T + \underline{X} \underline{Y}^T}{r} \right] - \frac{\mu}{r^3} \Gamma \left[I - \frac{\underline{X} \underline{X}^T}{r^2} \right] + \frac{\mu}{r^3} H \left[\frac{3x_3}{r^3} \underline{X} \underline{X}^T - \frac{1}{r} (\underline{X} \underline{a}^T + \underline{a} \underline{X}^T) - \frac{x_3}{r} I \right] \\ & + \frac{\mu}{r^3} \begin{bmatrix} N & -\Omega & 0 \\ -\Omega & -N & 0 \\ 0 & 0 & 0 \end{bmatrix} \end{aligned} \quad (8)$$

Collecting like terms, we get

$$\begin{aligned} \frac{\partial^2 U}{\partial \underline{X}^2} = & \frac{\mu}{r^3} \left(L + M \frac{x_3^2}{r^2} + 2P \frac{x_3}{r} + \Gamma + 3H \frac{x_3}{r} \right) \frac{\underline{X} \underline{X}^T}{r^2} \\ & - \frac{\mu}{r^3} \left(M \frac{x_3}{r} + P + H \right) \left[\frac{\underline{X} \underline{a}^T + \underline{a} \underline{X}^T}{r} \right] \\ & + \frac{\mu}{r^3} M \underline{a} \underline{a}^T - \frac{\mu}{r^3} \left(\Gamma + \frac{x_3}{r} H \right) I \\ & + \frac{\mu}{r^3} \left\{ (\underline{a} \underline{a}^T + \underline{a} \underline{a}^T) - \frac{x_3}{r^2} (\underline{a} \underline{X}^T + \underline{X} \underline{a}^T) - \left(\frac{\underline{Y} \underline{X}^T + \underline{X} \underline{Y}^T}{r} \right) + \begin{bmatrix} N & -\Omega & 0 \\ -\Omega & -N & 0 \\ 0 & 0 & 0 \end{bmatrix} \right\} \end{aligned} \quad (9)$$

Recalling that

$$\Lambda = (\Gamma + \varepsilon H)$$

$$\hat{\underline{X}} = \frac{\underline{X}}{r}$$

and defining

$$F \equiv L + \varepsilon(M\varepsilon + 2(P + H)) + \Lambda$$

$$G \equiv -(M\varepsilon + P + H)$$

$$\underline{d} \equiv \varepsilon \underline{a} + \underline{y}$$

we have finally

$$\begin{aligned} \frac{\partial^2 U}{\partial \underline{X}^2} = & \frac{\mu}{r^3} \left\{ [\hat{X} \mid \underline{a}] \begin{bmatrix} F & G \\ G & M \end{bmatrix} \begin{bmatrix} \hat{X}^T \\ \underline{a}^T \end{bmatrix} \right. \\ & \left. + [\hat{X} \mid \underline{d}] \begin{bmatrix} 0 & -1 \\ -1 & 0 \end{bmatrix} \begin{bmatrix} \hat{X}^T \\ \underline{d}^T \end{bmatrix} + \begin{bmatrix} N - \Lambda & -\Omega & Q \\ -\Omega & -(N + \Lambda) & R \\ Q & R & -\Lambda \end{bmatrix} \right\} \end{aligned} \quad (10)$$

Note that the final result for the second partial derivative of the geopotential is a rather simple, compact, symmetric matrix equation.

SECTION 5

NUMERICAL VERIFICATION

The equations in this report were coded and checked numerically using GOTPOT.

The computer program, GOTPOT, is given in the appendix in FORTRAN. Also included in the appendix is the truncated set of coefficients used in the numerical verification. The coefficients are contained in the subroutine, GEM10, which must be called before GOTPOT is called.

Two types of numerical checks were done. The first derives from the fact that U satisfies Laplace's equation; i.e., that

$$\frac{\partial^2 U}{\partial x_1^2} + \frac{\partial^2 U}{\partial x_2^2} + \frac{\partial^2 U}{\partial x_3^2} = 0$$

Therefore, the trace of $\frac{\partial^2 U}{\partial \underline{X}^2}$ must be zero.

Secondly, numerical partials of g were taken by varying x_1 , x_2 , and $x_3 \pm 4$ meters. The central difference derivatives were subtracted from the appropriate term in $\partial^2 U / \partial \underline{X}^2$ and then normalized by dividing the error by that term. The results of the numerical tests with $n = m = 4$ and $n = m = 5$ are given in table I.

Note that the trace is zero to 16 decimal digits compared to the individual diagonal elements. This is as close to zero as double precision arithmetic can get.

Note also that the numerical partials agree with the analytic to at least nine digits. These are inherently not as accurate as the trace since we are comparing a numerical partial to an analytic one. Nonetheless, the agreement is outstanding.

It should be noted that different computers may get slightly different results for the trace and error terms because we are dealing with numbers at the edge of computer accuracy.

These tests, together with comparisons of the formulae with direct differentiation of U considering only J_2 (not presented here) lead to the following.

SECTION 6
CONCLUSIONS

A non-singular, recursive algorithm for computing the first and a symmetric, non-singular, recursive algorithm for computing the second derivative of the geopotential has been derived, programmed, and verified.

SECTION 7
REFERENCE

1. Mueller, Alan C. "A Fast Recursive Algorithm for Calculating the Forces Due to the Geopotential (Program: GEOPOT)." JSC Internal Note 75-FM-42, June 1975.

TABLE I.- NUMERICAL VERIFICATION OF RESULTS

NOMINAL \underline{X} = 5489150.0 , 802222.0 , 3140916.0 (meters)

ANALYTIC DGD \underline{X} $n = 4, m = 4$

0.18777919E-05	0.49927074E-06	0.19651588E-05
0.49927074E-06	-0.14652000E-05	0.28721411E-06
0.19651588E-05	0.28721411E-06	-0.41259196E-06

TRACE 5.29395592033937712E-23

NUMERICAL DGD \underline{X} $n = 4, m = 4$ ($dx = \pm 4$ meters)

0.18777919E-05	0.49927074E-06	0.19651588E-05
0.49927074E-06	-0.14652000E-05	0.28721411E-06
0.19651588E-05	0.28721411E-06	-0.41259196E-06

(A - N)/A

-0.39090978E-10	-0.30118256E-10	-0.44035904E-10
-0.25248719E-09	0.97483656E-11	0.45181657E-10
0.68954763E-10	-0.51455571E-10	-0.14525828E-09

ANALYTIC DGD \underline{X} $n = 5, m = 5$

0.18777323E-05	0.49925937E-06	0.19650747E-05
0.49925937E-06	-0.14651356E-05	0.28720884E-06
0.19650747E-05	0.28720884E-06	-0.41259666E-06

TRACE 4.76456032830543941E-22

NUMERICAL DGD \underline{X} $n = 5, m = 5$ ($dx = \pm 4$ meters)

0.18777323E-05	0.49925938E-06	0.19650747E-05
0.49925937E-06	-0.14651356E-05	0.28720884E-06
0.19650747E-05	0.28720884E-06	-0.41259666E-06

(A - N)/A

-0.28756763E-10	-0.18302270E-10	0.20757122E-10
0.48203922E-09	-0.75195398E-11	-0.16305638E-09
0.20757122E-10	-0.66417378E-10	0.23218316E-09

APPENDIX
COMPUTER LISTINGS

```

SUBROUTINE GOTPOT(X,G,DGDX,NAX,MAX,ID)
C
C  PROGRAMMED BY ROBERT G. GOTTLIEB
C
C  NOTE: ID = 2 FOR SECOND PARTIALS, MUST = 1 OTHERWISE
C
COMMON/GRCOEF/C(20),S(20),MU,RE
IMPLICIT DOUBLE PRECISION (A-H,O-Z)
DIMENSION X(3),G(3),DGDX(3,3),PN(0:20),PNM1(0:20),PNM2(0:20)
DIMENSION CTIL(0:20),STIL(0:20)
DOUBLE PRECISION MU,MUOR2,MUOR3,MXPNM,LAMBDA
R = DSQRT( X(1)**2 + X(2)**2 + X(3)**2 )
RI = 1.D0/R
XOR = X(1)*RI
YOR = X(2)*RI
ZOR = X(3)*RI
EP = ZOR
REOR = RE*RI
REORN = REOR
MUOR2 = MU*RI*RI
K = 0
DO 10 M = 1,MAX + ID
  PNM2(M) = 0.D0
10 PNM1(M) = 0.D0
  PNM2(0) = 1.D0
  PNM1(0) = EP
  PNM1(1) = 1.D0
  CTIL(0) = 1.D0
  STIL(0) = 0.D0
  CTIL(1) = XOR
  STIL(1) = YOR
  SUMH = 0.D0
  SUMGM = 1.D0
  SUMJ = 0.D0
  SUMK = 0.D0
  IF(ID.EQ.2)THEN
    SUML = 2.D0
    SUMM = 0.D0
    SUMN = 0.D0
    SUMO = 0.D0
    SUMP = 0.D0
    SUMQ = 0.D0
    SUMR = 0.D0
    SUMS = 0.D0
    SUMT = 0.D0
  ENDIF
C
DO 50 N = 2,NAX

```

```

REORN = REORN*REOR
N2M1 = N + N-1
NM1 = N-1
NP1 = N + 1
C ***** RECURSIVE COMPUTATION OF LEGENDRE POLYNOMIALS - PN(0)
C ***** AND FIRST ASSOCIATED LEGENDRE FUNCTIONS PN(1), PN (2)
PN(0) = (N2M1*EP*PNM1(0)-NM1*PNM2(0))/N
PN(1) = PNM2(1) + N2M1*PNM1(0)
PN(2) = PNM2(2) + N2M1*PNM1(1)
K = K + 1
SUMHN = PN(1)*C(K)
SUMGMN = PN(0)*C(K)*NP1
IF(ID.EQ.2)THEN
  SUMLN = SUMGMN*(NP1 + 1)
  SUMMN = PN(2)*C(K)
  SUMPN = SUMHN*NP1
ENDIF
IF(MAX.GT.0) THEN
  SUMJN = 0.D0
  SUMKN = 0.D0
  IF(ID.EQ.2)THEN
    SUMNN = 0.D0
    SUMON = 0.D0
    SUMQN = 0.D0
    SUMRN = 0.D0
    SUMSN = 0.D0
    SUMTN = 0.D0
  ENDIF
  CTIL(N) = CTIL(1)*CTIL(NM1)-STIL(1)*STIL(NM1)
  STIL(N) = STIL(1)*CTIL(NM1) + CTIL(1)*STIL(NM1)
  IF(N.LT.MAX)THEN
    LIM = N
  ELSE
    LIM = MAX
  ENDIF
  DO 20 M = 1,LIM
    MM1 = M-1
    MP1 = M + 1
C ***** RECURSIVE COMPUTATION OF ASSOCIATED LEGENDRE FUNCTIONS - PN(M + 1)
    PN(MP1) = PNM2(MP1) + N2M1*PNM1(M)
    MXPNM = M*PN(M)
    NM = K + M
    BNMTIL = C(NM)*CTIL(M ) + S(NM)*STIL(M )
    BNMTM1 = C(NM)*CTIL(MM1) + S(NM)*STIL(MM1)
    ANMTM1 = C(NM)*STIL(MM1)-S(NM)*CTIL(MM1)
    SUMHN = SUMHN + PN(MP1)*BNMTIL
    SUMGMN = SUMGMN + (N + MP1)*PN(M)*BNMTIL
    SUMJN = SUMJN + MXPNM*BNMTM1
    SUMKN = SUMKN-MXPNM*ANMTM1
    IF(ID.EQ.2)THEN
      MP2 = M + 2
      NPMP1 = N + MP1

```

```

PN(MP2) = PNM2(MP2) + N2M1*PNM1(MP1)
SUMLN = SUMLN + NPMP1*(MP1 + NP1)*PN(M)*BNMTIL
SUMMN = SUMMN + PN(MP2)*BNMTIL
SUMPN = SUMPN + NPMP1*PN(MP1)*BNMTIL
SUMQN = SUMQN + M*PN(MP1)*BNMTM1
SUMRN = SUMRN - M*PN(MP1)*ANMTM1
SUMSN = SUMSN + NPMP1*MXPNM*BNMTM1
SUMTN = SUMTN - NPMP1*MXPNM*ANMTM1
IF(M.GE.2)THEN
  MM2 = M-2
  SUMNN = SUMNN + MM1*MXPNM*(C(NM)*CTIL(MM2) + S(NM)*STIL(MM2))
  SUMON = SUMON + MM1*MXPNM*(C(NM)*STIL(MM2) - S(NM)*CTIL(MM2))
END IF
END IF
20 CONTINUE
SUMJ = SUMJ + REORN*SUMJN
SUMK = SUMK + REORN*SUMKN
IF(ID.EQ.2)THEN
  SUMN = SUMN + REORN*SUMNN
  SUMO = SUMO + REORN*SUMON
  SUMQ = SUMQ + REORN*SUMQN
  SUMR = SUMR + REORN*SUMRN
  SUMS = SUMS + REORN*SUMSN
  SUMT = SUMT + REORN*SUMTN
ENDIF
ENDIF
C ***** SUMS BELOW HERE HAVE VALUES WHEN M = 0
SUMH = SUMH + REORN*SUMHN
SUMGM = SUMGM + REORN*SUMGMN
IF(ID.EQ.2)THEN
  SUML = SUML + REORN*SUMLN
  SUMM = SUMM + REORN*SUMMN
  SUMP = SUMP + REORN*SUMPN
ENDIF
K = K + N
C ***** SHIFT LEGENDRE POLYNOMIALS DOWN
IF(N.LT.NAX)THEN
  DO 40 J = 0,N
    PNM2(J) = PNM1(J)
    PNM1(J) = PN(J)
  40 CONTINUE
ENDIF
50 CONTINUE
C
LAMBDA = SUMGM + EP*SUMH
G(1) = -MUOR2*(LAMBDA*XOR-SUMJ)
G(2) = -MUOR2*(LAMBDA*YOR-SUMK)
G(3) = -MUOR2*(LAMBDA*ZOR-SUMH)
IF(ID.EQ.2)THEN
C ***** NEED TO CONSTRUCT 2ND PARTIAL DERIVATIVE MATRIX
GG = -(SUMM*EP + SUMP + SUMH)
FF = SUML + LAMBDA + EP*(SUMP + SUMH - GG)

```



```
D1 = EP*SUMQ + SUMS
D2 = EP*SUMR + SUMT
MUOR3 = MUOR2*RI
DGDx(1,1) = MUOR3*((FF*XOR-2.D0*D1)*XOR-LAMBDA + SUMN)
DGDx(2,2) = MUOR3*((FF*YOR-2.D0*D2)*YOR-LAMBDA-SUMN)
DGDx(3,3) = MUOR3*((FF*ZOR + 2.D0*GG)*ZOR-LAMBDA + SUMM)
DGDx(1,2) = MUOR3*((FF*YOR-D2)*XOR-D1*YOR-SUMO)
DGDx(1,3) = MUOR3*((FF*XOR-D1)*ZOR + GG*XOR + SUMQ)
DGDx(2,3) = MUOR3*((FF*YOR-D2)*ZOR + GG*YOR + SUMR)
DGDx(2,1) = DGDx(1,2)
DGDx(3,1) = DGDx(1,3)
DGDx(3,2) = DGDx(2,3)
ENDIF
RETURN
END
```

SUBROUTINE GEM10

```
COMMON/GRCOEF/C(20),S(20),MU,RE
DOUBLE PRECISION MU,CC(5,0:5),SS(5,0:5)
IMPLICIT DOUBLE PRECISION (A-H,O-Z)
```

- * The constants given herein are taken from
- *
- * "Gravity Model Improvement Using Geos 3 (GEM 9 and GEM 10)"
- * by
- * F. Lerch, S. Klosko, R. Laubscher, and C. Wagner
- * Journal of Geophysical Research, Vol. 84, NO. B8, July 30, 1979

- * From Table 8 --Comparison of the IAG 1975 and GEM Geodetic Parameters
- MU = 39860047.0D7
- RE = 6378139.0D0

- * From Table 4--Goddard Earth Model 10 Normalized Coefficients
- * (We use an abbreviated set for reasons of space)
- CC(2, 0) = -484.16544D0
- CC(3, 0) = 0.95838D0
- CC(4, 0) = 0.54112D0
- CC(5, 0) = 0.06862D0

- CC(2, 1) = 0.00104D0
- SS(2, 1) = -0.00243D0
- CC(3, 1) = 2.02855D0
- SS(3, 1) = 0.25197D0
- CC(4, 1) = -0.53521D0
- SS(4, 1) = -0.46926D0
- CC(5, 1) = -0.05117D0
- SS(5, 1) = -0.09379D0

- CC(2, 2) = 2.43404D0
- SS(2, 2) = -1.39907D0
- CC(3, 2) = 0.89272D0
- SS(3, 2) = -0.62346D0
- CC(4, 2) = 0.35208D0
- SS(4, 2) = 0.66404D0
- CC(5, 2) = 0.65146D0
- SS(5, 2) = -0.32769D0

- CC(3, 3) = 0.70028D0
- SS(3, 3) = 1.41250D0
- CC(4, 3) = 0.98850D0
- SS(4, 3) = -0.20179D0
- CC(5, 3) = -0.46712D0
- SS(5, 3) = -0.20298D0

- CC(4, 4) = -0.19531D0
- SS(4, 4) = 0.29883D0
- CC(5, 4) = -0.28754D0

SS(5, 4) = 0.04990D0

CC(5, 5) = 0.15617D0

SS(5, 5) = -0.65983D0

- * The unnormalized coefficients, in single subscript form, used in GOTPOT
- * are computed below.

```

DO 200 N = 2,5
DO 100 M = 0,N
J = (N**2 + N -4)/2 + M
IF( M .EQ. 0 ) THEN
  A = 2*N + 1
  C(J) = DSQRT(A)*CC(N,0)*1.0D-6
  S(J) = 0.D0
ELSE
  COEF = 2*(2*N + 1)*FACTNK(N-M,N + M)
  COEF = DSQRT(COEF)*1.0D-6
  C(J) = COEF * CC(N,M)
  S(J) = COEF * SS(N,M)
END IF
100 CONTINUE
200 CONTINUE
RETURN
END

```

```

DOUBLE PRECISION FUNCTION FACTNK(N,K)
DOUBLE PRECISION ANSWER
* This routine computes n!/k!
INTEGER START,STOP
IF(N.GT.K)THEN
  START = K + 1
  STOP = N
ELSE IF(N.EQ.K)THEN
  FACTNK = 1.D0
  RETURN
ELSE
  START = N + 1
  STOP = K
END IF
ANSWER = START
DO 100 I = START + 1,STOP
  ANSWER = ANSWER*I
100 CONTINUE
IF(K.GT.N)ANSWER = 1.D0/ANSWER
FACTNK = ANSWER
RETURN
END

```

DISTRIBUTION FOR JSC IN 89-FM-10

NASA - JSC

DM/F. Lowes

D. Pearson

EH2/W. L. Jackson (2)

EH5/J. Yeo

J. Thibodeau

EG/A. Bordano

EG4/J. Blucker

J. Currie

C. Bean

W. Lear

D. Geller

R. Eckelkamp

I. Saulietis

ET/C. Graves

ET3/H. Beck

M. Jenness

S. Lewis

ET4/D. Long

D. Hoffman

ET14/B. Cockrell

IZ/E. Lineberry

JM2/Technical Library (3)

JM21/L. Reed (2)

JM88/Distribution (15 + extras)

PT4/R. Lea

CSC - Houston

M30/G. Brown

CSDL - Houston

G. Mueller

IBM - Houston

O. Graf

G. Smith

MDSSC -ES - Houston

R. Pritchard

D. Mulcihy

S. Sangchat

R. Gottlieb (20)

M. Fraietta

V. Holloman

B. McCormick

G. Hirsch

W. Greenwell

D. Hull

V. Bond (20)

L. Zyla

J. Compton

S. Zobal

J. Vedder

D. Pirker

L. Guderian

R. Carmody

C. Semar

M. Lardas

G. Lardas

Y. Park

L. Rater

MSDDC-SSD - Houston

H. Stegall

M. Little

R. Corson

W. Long

OMNIPLAN - Houston

Editing

17226 Mercury

Houston, TX 77058

RSOC - Houston

D. Adamo

University of Texas at Austin

Department of Aerospace Engineering

Austin, TX 78712

Attn: V. Szebehely

R. Duncombe

W. Fowler